
1 Syntax

We use an extended BNF grammar to define the syntax. The general form of a production is

nonterminal ::= alternative | alternative | · · · | alternative

The following extensions are used:

a · · · a list of one or more as.
a, · · · a list of one or more as separated by commas:

“a” or “a, a” or “a, a, a”, etc.
{ a } { and } are meta-syntactical brackets

treating a as one syntactic category.
[ a ] an optional a: “ ” or “a”.

Nonterminal symbols appear in italic face. Terminal symbols appear in the face like this: “terminal”, and may be
surrounded by “ and ” for emphasis or to avoid confusion with meta characters used in the extended BNF. We will
refer terminal symbols other than self-terminating characters (see section 2.2) as keywords in this document.

*sawada@cafeobj.org

1.1 CafeOBJ Spec

spec ::= { module | view | eval } · · ·

A CafeOBJ spec is a sequence of module (module declaration – section 1.2), view (view declaration – section 1.4)
or eval (reduce or execute term – section 1.5).

1.2 Module Declaration

module ::= module_type module_name [ parameters ] [ principal_sort ]
“{” module_elt · · · “}”

module_type ::= module | module! | module*
module_name ::= ident – ¹

parameters ::= “(” parameter, · · · “)”
parameter ::= [ protecting | extending | including ] paramter_name :: module_expr – ²³
parameter_name ::= ident

principal_sort ::= principal-sort sort_name
module_elt ::= import | sort | operator | variable | axiom | macro | comment – ⁴
import ::= { protecting | extending | including | using }“(” module_expr “)”
sort ::= visible_sort | hidden_sort
visible_sort ::= “[” sort_decl, · · · “]”
hidden_sort ::= “*[” sort_decl, · · · “]*”
sort_decl ::= sort_name · · · [ supersorts · · · ]
supersorts ::= < sort_name · · ·
sort_name ::= sort_symbol[ qualifier ] – ⁵
sort_symbol ::= ident
qualifier ::= “.”module_name

operator ::= { op | bop } operator_symbol : [ arity ] -> coarity [ op_attrs ] – ⁶
arity ::= sort_name · · ·
coarity ::= sort_name
op_attrs ::= “{” op_attr · · · “}”
op_attr ::= constr | associative | commutative | idempotent | { id: | idr: }“(” term “)”

| strat: “(” natural · · · “)” | prec: natural | l-assoc | r-assoc | coherent | demod – ⁷
variable ::= var var_name : sort_name | vars var_name · · · : sort_name
var_name ::= ident

axiom ::= equation | cequation | transition | ctransition | fol
equation ::= { eq | beq } [ label ] term = term “.”
cequation ::= { ceq | bceq } [ label ] term = term if term “.”
transition ::= { trans | btrans } [ label ] term => term “.”
ctransition ::= { ctrans | bctrans } [ label ] term => term if term “.”
fol ::= ax[ label ] term “.”
label ::= “[” ident · · · “]:”

macro ::= #define term ::= term “.”
¹The nonterminal ident is for identifiers and will be defined in the section 2.3.
²module_expr is defined in the section 1.3.
³If optional [ protecting | extending | including ] is omitted, it is defaulted to protecting.
⁴comment is descussed in section 2.5.
⁵There must not be any separators (see section 2) between ident and qualifier.
⁶operator_symbol is defined in section 2.4.
⁷natural is a natural number written in ordinal arabic notation.

1.3 Module Expression

module_expr ::= module_name | sum | rename | instantiation | “(”module_expr“)”
sum ::= module_expr { + module_expr } · · ·
rename ::= module_expr * “{”rename_map, · · · “}”
instantiation ::= module_expr “(”{ ident[qualifier] <= aview}, · · · “)”
rename_map ::= sort_map | op_map
sort_map ::= { sort | hsort } sort_name -> ident
op_map ::= { op | bop } op_name -> operaotr_symbol
op_name ::= operator_symbol | “(”operator_symbol“)”qualifier
aview ::= view_name | module_expr

| view to module_expr “{”view_elt, · · · “}”
view_name ::= ident
view_elt ::= sort_map | op_view | variable
op_view ::= op_map | term -> term

When a module expression is not fully parenthesized, the proper nesting of subexpressions may be ambiguous. The
following precedence rule is used to resolve such ambiguity:

sum < rename < instantiation

1.4 View Declaration

view ::= view view_name from module_expr to module_expr
“{” view_elt, · · · “}”

1.5 Evaluation

eval ::= { reduce | behavioural-reduce | execute } context term “.”
context ::= in module_expr :

The interpreter has a notion of current module which is specified by a module_expr and establishes a context. If it is
set, context can be omitted.

1.6 Sugars and Abbriviations

Module type There are following abbreviations for module_type.

Keyword Abbriviation
module mod
module! mod!
module* mod*

Module Declaration

make ::= make module_name “(” module_expr “)”

make is a short hand for declaring module of name module_name which imports module_expr with protecting mode.
make FOO (BAR * {sort Bar -> Foo})
is equivalent to
module FOO { protecting (BAR * {sort Bar -> Foo}) }

Principal Sort principal-sort can be abbriviated to psort.

Import Mode For import modes, the following abbriviations can be used:

Keyword Abbriviation
protecting pr
extending ex
including inc
using us

Simultaneous Operator Declaration Several operators with the same arity, coarity and operator attributes can be
declared at once by ops. The form

ops operator_symbol1 · · · operator_symboln : arity -> coarity op_attrs

is just equivalent to the following multiple operator declarations:

op operator_symbol1 : arity -> coarity op_attrs
...

op operator_symboln : arity -> coarity op_attrs

bops is the counterpart of ops for behavioural operators.

bops operator_symbol · · · : arity -> coarity op_attrs

In simultaneous declarations, parentheses are sometimes necessary to separate operator symbols. This is always
required if an operator symbol contains dots, blank characters or underscores.

Predicate Predicate declaration (predicate) is a syntactic sugar for declaring Bool valued operators, and has the
syntax:

predicate ::= pred operator_symbol : arity [ op_attrs ] – ⁸

The form

pred operator_symbol : arity op_attrs

is equivalent to:

op operator_symbol : arity -> Bool op_attrs

Operator Attributes The following abbriviations are available:

Keyword Abbriviation
associative assoc
commutative comm
idempotent idem

⁸You cannot use sort_name of the same character sequence as that of any keywords, i.e., module, op, vars, etc. in arity.

Axioms For the keywords introducing axioms, the following abbriviations can be used:

Keyword Abbriviation Keyword Abbriviation
ceq cq bceq bcq
trans trns ctrans ctrns
btrans btrns bctrans bctrns

Blocks of Declarations References to (importations of) other modules, signature definitions and axioms can be
clusterd in blocked declarations:

imports ::= imports “{”
{ import | comment } · · ·

“}”
signature ::= signature “{”

{ sort | record | operator | comment } · · ·
“}”

axioms ::= axioms “{”
{ variable | axiom | comment } · · ·

“}”

Views To reduce the complexity of views appearing in module instantiation, some sugars are provided.
First, it is possible to identify parameters by positions, not by names. For example, if a parameterized module is
declared like
module! FOO (A1 :: TH1, A2 :: TH2) { ... }
the form
FOO(V1, V2)
is equivalent to
FOO(A1 <= V1, A2 <= V2)
Secondly, view to construct in arguments of module instantiations can always be omitted. That is,
FOO(A1 <= view to module_expr{...})
can be written as
FOO(A1 <= module_expr{...})

Evaluation

Keyword Abbriviation
reduce red
bereduce bred
execute exec

2 Lexical Considerations

ACafeOBJ spec is written as a sequence of tokens and separators. A token is a sequence of “printing”ASCII characters
(octal 40 through 176).⁹ A separator is a “blank” character (space, vertical tab, horizontal tab, carriage return, newline,
form feed). In general, any mumber of separators may appear between tokens.

⁹The current interpreter accepts Unicode characters also, but this is beyond the definition of CafeOBJ language.

2.1 Reserved Word

There are no reserved word in CafeOBJ. One can use keywords such asmodule, op, var, or signature, etc. for identifiers
or operator symbols.

2.2 Self-terminating Characters

The following eight characters are always treated as self-terminating, i.e., the character itself construct a token.

( ) , [ ] { } ;

2.3 Identifier

Nonterminal ident is for identifier which is a sequnce of any printing ASCII characters except the followings:

self-terminating characters (see section 2.2)
. (dot)
” (double quote)

Upper- and lowercase are distinguished in identifiers. idents are used for module names (module_name), view names
(view_name), parameter names (parameter_name), sort symbols (sort_symbol), variables(var_name), slot names
(slot_name) and labels (label).

2.4 Operator Symbol

The nonterminal operator_symbol is used for naming operators (operator) and is a sequence of any ASCII characters
(self-terminating characters or non-printing characters can be an element of operator names.)¹⁰
Underscores are specially treated when they apper as a part of operator names; they reserve the places where arguments
of the operator are inserted. Thus the single underscore cannot be a name of an operator.

2.5 Comments and Separators

A comment is a sequence of characters that begins with one of the following four character sequences

-- -->
** **>

which ends with a newline character, and contains only printing ASCII characters and horizontal tabs in between.
A separator is a blank character (space, vertical tab, horizontal tab, carriage return, newline, from feed). One or more
separators must appear between any two adjuacent non-self-terminating tokens.¹¹
Comments also act as separators, but their apperance is limited to some specific places (see section 1).

¹⁰The current implementation does not allow EOT character (control-D) to be an element of operator symbol.
¹¹The same rule is applied to term. Further, if an operator_symbol contains blanks or self-terminating characters, it is sometimes neccessary to

enclose a term with such operator as top by parentheses for disambiguation.

Multiline comments A multiple lines which starts with #| and ends with |# is treated as multiline comment.

#|---------------------------------------------------
This is an example of multiline comment.
Multiline comments are used for large text descriptions of
code or to comment out chunks of code while developping
your specification.
Multiline comments are ignored by the system.

---------------------------------------------------|#

